

Human Lassa Case Rate

This code shows the analyses performed on the system of ordinary differential equations that describe the spillover of Lassa virus into humans.

■ Define model

Define the system of ODEs that describe human infection.

```
dsdt = b - d s - F s + λ r;
didt = F s - (d + γ) i;
drdt = (1 - μ) γ i - d r - λ r;
```

■ Derive relationship between seroprevalence and new case rate

Find the equilibrium.

```
eq = FullSimplify[Solve[{dsdt == 0, didt == 0, drdt == 0}, {s, i, r}]]
```

$$\left\{ \left\{ s \rightarrow \frac{b (d + \gamma) (d + \lambda)}{d (d + F) (d + \gamma) + d (d + F + \gamma) \lambda + F \gamma \lambda \mu}, \right. \right.$$

$$\left. i \rightarrow \frac{b F (d + \lambda)}{d (d + F) (d + \gamma) + d (d + F + \gamma) \lambda + F \gamma \lambda \mu}, r \rightarrow -\frac{b F \gamma (-1 + \mu)}{d (d + F) (d + \gamma) + d (d + F + \gamma) \lambda + F \gamma \lambda \mu} \right\}$$

Find seroprevalence at equilibrium (this is the data we have to work with). Note that this requires calculating the steady state population size in the presence of LASV virulence.

```
Pop = FullSimplify[(s + i + r) /. eq]
```

$$\left\{ \frac{b (d^2 + \gamma \lambda + d (F + \gamma + \lambda) + F (\gamma + \lambda - \gamma \mu))}{d (d + F) (d + \gamma) + d (d + F + \gamma) \lambda + F \gamma \lambda \mu} \right\}$$

Variable [count] refers to the population size in a pixel of the population raster. Solve for the birth rate [b] for a given value of [count].

```
bsolve = Solve[Pop == count, b]
```

$$\left\{ \left\{ b \rightarrow \frac{\text{count} (d^3 + d^2 F + d^2 \gamma + d F \gamma + d^2 \lambda + d F \lambda + d \gamma \lambda + F \gamma \lambda \mu)}{(d^2 + d F + d \gamma + F \gamma + d \lambda + F \lambda + \gamma \lambda - F \gamma \mu)} \right\} \right\}$$

Calculate seroprevalence in terms of model parameters.

```
Rstar = FullSimplify[r / (Pop) /. eq] // Flatten
```

$$\left\{ -\frac{F \gamma (-1 + \mu)}{d^2 + \gamma \lambda + d (F + \gamma + \lambda) + F (\gamma + \lambda - \gamma \mu)} \right\}$$

Solve for the force of infection, F, in terms of seroprevalence.

```
foi = FullSimplify[Solve[hsero == Rstar, F]] // Flatten
```

$$\left\{ F \rightarrow -\frac{hsero (d + \gamma) (d + \lambda)}{d hsero + hsero \lambda + \gamma (-1 + hsero + \mu - hsero \mu)} \right\}$$

Rewrite the solution for [b], defined in [bsolve], with the force of infection terms F replaced with the expression above

```
bsolve = FullSimplify[bsolve /. foi]
```

$$\left\{ \left\{ b \rightarrow -\frac{\text{count} (d + d (-1 + hsero) \mu + hsero \lambda \mu)}{-1 + \mu} \right\} \right\}$$

The rate of new cases of Lassa is F*s. Calculate this, and rewrite everything in terms of seroprevalence.

```
NewCaseRate = FullSimplify[(F /. foi) * (s /. eq /. foi) /. bsolve] // Flatten
```

$$\left\{ -\frac{\text{count} hsero (d + \gamma) (d + \lambda)}{\gamma (-1 + \mu)} \right\}$$

■ Compare case rate estimates for virulence vs no virulence

Calculate the ratio of cases with, versus without, virulence.

```
ρ = FullSimplify[(NewCaseRate /. eq) / (NewCaseRate /. eq /. μ → 0)] // Flatten
```

$$\left\{ \frac{1}{1 - \mu} \right\}$$

In words, including virulence increases the estimated rate of LASV cases by $\rho = (1 - \mu)^{-1}$.

```
ρ /. μ → 0.02
```

```
{1.02041}
```

When $\mu = 0.02$ (lower value from McCormick, Webb, 1987), increase is ~1.02, meaning roughly 2% more cases occur.

■ Compare estimates for reinfection vs no reinfection

Calculate the ratio of cases with, versus without, reinfection.

```
FullSimplify[(NewCaseRate /. eq) / (NewCaseRate /. eq /. λ → 0)] // Flatten
```

$$\left\{ \frac{d + \lambda}{d} \right\}$$

Reinfection multiplies the case rate by $\frac{d+\lambda}{d}$. Calculate the effect of reinfection for a mean lifespan of 50 years, and reinfection rate $\lambda = 0.064$ / yr (from McCormick, Webb, 1987).

`Pars = {d → 0.02};`

Ratio of estimates of new cases with ($\lambda = 0.064$) and without ($\lambda = 0$) seroreversion is

`(NewCaseRate /. λ → 0.064) / (NewCaseRate /. λ → 0) /. Pars`
`{4.2}`

Including reinfection multiplies the case estimates by a factor of 420%; including virulence increases estimates by 2%.

Human LASV cases with undetectable antibody immune class

At the request of a reviewer, we also investigate the possibility that a fraction, a , of recovered individuals with immunity (R) transition into a class with undetectable immunity (C). This might be the case, for example, if immunity were caused primarily by a T-cell response. into humans.

■ Define model

Define the system of ODEs that describe human infection. Note the additional class, c , and the rate term a .

`dsdt = b - d s - F s + λ (1 - a) r;`
`didt = F s - (d + γ) i;`
`drdt = (1 - μ) γ i - d r - λ r;`
`dcdt = a λ a m r - d c;`

■ Derive relationship between seroprevalence and new case rate

The remainder of this analysis is identical to what was performed above. Note that the equilibrium values of s are smaller when $a > 0$. Find the equilibrium.

```
eq = FullSimplify[Solve[{dsdt == 0, didt == 0, drdt == 0, dcdt == 0}, {s, i, r, c}]]
```

$$\left\{ \left\{ \begin{aligned} s &\rightarrow \frac{b (d + \gamma) (d + \lambda)}{d^3 + d F \gamma + d (F + \gamma) \lambda + d^2 (F + \gamma + \lambda) + F \gamma \lambda (a + \mu - a \mu)}, \\ i &\rightarrow \frac{b F (d + \lambda)}{d^3 + d F \gamma + d (F + \gamma) \lambda + d^2 (F + \gamma + \lambda) + F \gamma \lambda (a + \mu - a \mu)}, \\ r &\rightarrow -\frac{b F \gamma (-1 + \mu)}{d^3 + d F \gamma + d (F + \gamma) \lambda + d^2 (F + \gamma + \lambda) + F \gamma \lambda (a + \mu - a \mu)}, \\ c &\rightarrow -\frac{a b F \lambda \mu \gamma (-1 + \mu)}{d (d^3 + d F \gamma + d (F + \gamma) \lambda + d^2 (F + \gamma + \lambda) + F \gamma \lambda (a + \mu - a \mu))} \end{aligned} \right\} \right\}$$

Find seroprevalence at equilibrium (this is the data we have to work with). Note that this requires calculating the steady state population size in the presence of LASV virulence.

```
Pop = FullSimplify[(s + i + r + c) /. eq]
```

$$\left\{ \left(\frac{b (d^3 + d \gamma \lambda + d^2 (F + \gamma + \lambda) - a F \lambda \mu \gamma (-1 + \mu) + d F (\gamma + \lambda - \gamma \mu))}{d (d^3 + d F \gamma + d (F + \gamma) \lambda + d^2 (F + \gamma + \lambda) + F \gamma \lambda (a + \mu - a \mu))} \right) \right\}$$

Variable [count] refers to the population size in a pixel of the population raster. Solve for the birth rate [b] for a given value of [count].

```
bsolve = Solve[Pop == count, b]
```

$$\left\{ \left\{ b \rightarrow \frac{\text{count } d (d^3 + d^2 F + d^2 \gamma + d F \gamma + d^2 \lambda + d F \lambda + d \gamma \lambda + a F \gamma \lambda + F \gamma \lambda \mu - a F \gamma \lambda \mu)}{(d^3 + d^2 F + d^2 \gamma + d F \gamma + a F \lambda \mu \gamma + d^2 \lambda + d F \lambda + d \gamma \lambda - d F \gamma \mu - a F \lambda \mu \gamma)} \right\} \right\}$$

Calculate seroprevalence in terms of model parameters.

```
Rstar = FullSimplify[r / (Pop) /. eq] // Flatten
```

$$\left\{ -\frac{d F \gamma (-1 + \mu)}{d^3 + d \gamma \lambda + d^2 (F + \gamma + \lambda) - a F \lambda \mu \gamma (-1 + \mu) + d F (\gamma + \lambda - \gamma \mu)} \right\}$$

Solve for the force of infection, F, in terms of seroprevalence. Note that the equilibrium value of foi is made larger by the parameter a.

```
foi = FullSimplify[Solve[hsero == Rstar, F] // Flatten]
```

$$\left\{ F \rightarrow -\frac{(d \text{ hsero } (d + \gamma) (d + \lambda))}{(d^2 \text{ hsero } + d \text{ hsero } \lambda - a \text{ hsero } \lambda \mu \gamma (-1 + \mu) + d \gamma (-1 + \text{ hsero } + \mu - \text{ hsero } \mu))} \right\}$$

Rewrite the solution for [b], defined in [bsolve], with the force of infection terms F replaced with the expression above

```
bsolve = FullSimplify[bsolve /. foi]
```

$$\left\{ \left\{ b \rightarrow -\frac{1}{-1 + \mu} \text{count } (d + a \text{ hsero } (\lambda \mu - \lambda) (-1 + \mu) + d (-1 + \text{ hsero } + \mu - \text{ hsero } \mu)) \right\} \right\}$$

The rate of new cases of Lassa is F*s. Calculate this, and rewrite everything in terms of seroprevalence. The decrease in s at equilibrium that is caused by a is exactly cancelled by the increase in force of

infection that is caused by a.

NewCaseRate = FullSimplify[(F /. foi) * (s /. eq /. foi) /. bsolve] // Flatten

$$\left\{ -\frac{\text{count hsero} (d + \gamma) (d + \lambda)}{\gamma (-1 + \mu)} \right\}$$